
	*****************************************************************
	* Taxing the Poor Twice: Bandwidth and the Value of Consumption * 
	*****************************************************************
	
	* PAPER TABLES 
		
	
	*1. Importing Merged Data 

	use "$dir_data/bandwidth_long_final.dta" , clear


	
	
	* Table S1: Baseline Summary Statistics

	* Counting observations
	count if treat==0 & tag_pid ==1
	local N_control = string(r(N), "%3.0f")
	count if treat==1 & tag_pid==1
	local N_priming = string(r(N), "%3.0f")
	count if treat==2 & tag_pid==1
	local N_memory = string(r(N), "%3.0f")
	count if treat==3 & tag_pid==1
	local N_thirst = string(r(N), "%3.0f")
	

	* Demographics 
	local varlist "age female married hh_size tamil_read educ_upto employed_outhh savings_cash earn_daily_hh_p95"
	foreach v in `varlist' {
		forvalues n = 0(1)3 {
			sum `v' if treat == `n' & tag_pid ==1
				local mean_`v'_t`n' = string(round(r(mean), .01),"%3.2f")
				local sd_`v'_t`n' = string(round(r(sd), .01),"%3.2f") 
				*string is just a trick to get rid of rounding problems (with unroundable numbers)
			if "`v'" == "savings_cash" | "`v'" == "earn_daily_hh"  {
				local mean_`v'_t`n' = substr("`mean_`v'_t`n''",1,strlen("`mean_`v'_t`n''")-3)
				display "`mean_`v'_t`n''"
				local sd_`v'_t`n' = substr("`sd_`v'_t`n''",1,strlen("`sd_`v'_t`n''")-3) 
				}
			}
		forvalues n = 1(1)3 {
			ttest `v' if ((treat == 0 | treat == `n') & tag_pid ==1), by(treat)
			local p_`v'_t`n' = string(round(r(p), .01),"%3.2f")
			}
		}	
		
	* Baseline Covariates 
	local varlist "stress_bs thirst_bs"
	foreach v in `varlist' {
		forvalues n = 0(1)3 {
			sum `v' if treat == `n' & tag_pid ==1
			local mean_`v'_t`n' = string(round(r(mean), .01),"%3.2f") 
			local sd_`v'_t`n' = string(round(r(sd), .01),"%3.2f")  
			*string is just a trick to get rid of rounding problems (with unroundable numbers)
			}
			
		forvalues n = 1(1)3 {
			ttest `v' if ((treat == 0 | treat == `n') & tag_pid ==1), by(treat)
			local p_`v'_t`n' = string(round(r(p), .01),"%3.2f")
			}
		}
		
		
	* Export Table
		
		cd "${dir_tables}"
		file open f using "tableS1_balance.tex", write replace
		file write f "\begin{tabular}{l*{8}{c}}" _n ///+
			"\toprule" _n ///
			" & (1) & (2) & (3) & (4) & (5) & (6) & (7) \\" _n ///
			"	&\multicolumn{1}{c}{Control}&\multicolumn{1}{c}{Financial Stress}&\multicolumn{1}{c}{Memory}&\multicolumn{1}{c}{Thirst}&\multicolumn{1}{c}{1 = 2}&\multicolumn{1}{c}{1 = 3}&\multicolumn{1}{c}{1 = 4}\\" _n ///
			"\midrule" _n ///
			"Age & `mean_age_t0' & `mean_age_t1' & `mean_age_t2' & `mean_age_t3' & `p_age_t1' & `p_age_t2' & `p_age_t3' \\" _n ///
			" & (`sd_age_t0') & (`sd_age_t1') & (`sd_age_t2') & (`sd_age_t3') & & & \\" _n ///	
			"Female & `mean_female_t0' & `mean_female_t1' & `mean_female_t2' & `mean_female_t3' & `p_female_t1' & `p_female_t2' & `p_female_t3' \\" _n ///
			" & (`sd_female_t0') & (`sd_female_t1') & (`sd_female_t2') & (`sd_female_t3') & & & \\" _n ///			
			"Married & `mean_married_t0' & `mean_married_t1' & `mean_married_t2' & `mean_married_t3' & `p_married_t1' & `p_married_t2' & `p_married_t3' \\" _n ///
			" & (`sd_married_t0') & (`sd_married_t1') & (`sd_married_t2') & (`sd_married_t3') & & & \\" _n ///	
			"Household size & `mean_hh_size_t0' & `mean_hh_size_t1' & `mean_hh_size_t2' & `mean_hh_size_t3' & `p_hh_size_t1' & `p_hh_size_t2' & `p_hh_size_t3' \\" _n ///
			" & (`sd_hh_size_t0') & (`sd_hh_size_t1') & (`sd_hh_size_t2') & (`sd_hh_size_t3') & & & \\" _n ///									
			"Literacy & `mean_tamil_read_t0' & `mean_tamil_read_t1' & `mean_tamil_read_t2' & `mean_tamil_read_t3' & `p_tamil_read_t1' & `p_tamil_read_t2' & `p_tamil_read_t3' \\" _n ///
			" & (`sd_tamil_read_t0') & (`sd_tamil_read_t1') & (`sd_tamil_read_t2') & (`sd_tamil_read_t3') & & & \\" _n ///														
			"Years of Education & `mean_educ_upto_t0' & `mean_educ_upto_t1' & `mean_educ_upto_t2' & `mean_educ_upto_t3' & `p_educ_upto_t1' & `p_educ_upto_t2' & `p_educ_upto_t3' \\" _n ///
			" & (`sd_educ_upto_t0') & (`sd_educ_upto_t1') & (`sd_educ_upto_t2') & (`sd_educ_upto_t3') & & & \\" _n ///				
			"Employed & `mean_employed_outhh_t0' & `mean_employed_outhh_t1' & `mean_employed_outhh_t2' & `mean_employed_outhh_t3' & `p_employed_outhh_t1' & `p_employed_outhh_t2' & `p_employed_outhh_t3' \\" _n ///
			" & (`sd_employed_outhh_t0') & (`sd_employed_outhh_t1') & (`sd_employed_outhh_t2') & (`sd_employed_outhh_t3') & & & \\" _n ///	
			"Savings (cash) & `mean_savings_cash_t0' & `mean_savings_cash_t1' & `mean_savings_cash_t2' & `mean_savings_cash_t3' & `p_savings_cash_t1' & `p_savings_cash_t2' & `p_savings_cash_t3' \\" _n ///
			" & (`sd_savings_cash_t0') & (`sd_savings_cash_t1') & (`sd_savings_cash_t2') & (`sd_savings_cash_t3') & & & \\" _n ///							
			"Mean daily hh earnings & `mean_earn_daily_hh_p95_t0' & `mean_earn_daily_hh_p95_t1' & `mean_earn_daily_hh_p95_t2' & `mean_earn_daily_hh_p95_t3' & `p_earn_daily_hh_p95_t1' & `p_earn_daily_hh_p95_t2' & `p_earn_daily_hh_p95_t3' \\" _n ///
			" & (`sd_earn_daily_hh_p95_t0') & (`sd_earn_daily_hh_p95_t1') & (`sd_earn_daily_hh_p95_t2') & (`sd_earn_daily_hh_p95_t3') & & & \\" _n ///	
			"\midrule" _n ///
			"Baseline Financial Stress & `mean_stress_bs_t0' & `mean_stress_bs_t1' & `mean_stress_bs_t2' & `mean_stress_bs_t3' & `p_stress_bs_t1' & `p_stress_bs_t2' & `p_stress_bs_t3' \\" _n ///
			" & (`sd_stress_bs_t0') & (`sd_stress_bs_t1') & (`sd_stress_bs_t2') & (`sd_stress_bs_t3') & & & \\" _n ///	
			"Baseline Thirst & `mean_thirst_bs_t0' & `mean_thirst_bs_t1' & `mean_thirst_bs_t2' & `mean_thirst_bs_t3' & `p_thirst_bs_t1' & `p_thirst_bs_t2' & `p_thirst_bs_t3' \\" _n ///
			" & (`sd_thirst_bs_t0') & (`sd_thirst_bs_t1') & (`sd_thirst_bs_t2') & (`sd_thirst_bs_t3') & & & \\" _n ///	
			"\midrule" _n ///
			"Number of Participants & `N_control' & `N_priming' & `N_memory' & `N_thirst'  & & & \\" _n ///
			"\bottomrule \\" _n ///
			"\end{tabular}" _n 
		file close f
		
		
		
		
		
		

		
		
	*Table S2 - Congnitive Performance  


	local cognitive_controls "age female tamil_write educ_upto"

	* Pooled Tests - pooled treated/not 
xi: reg cog_pay treated `cognitive_controls' if pid_count ==1 | pid_count ==2, cl(pid)  
	eststo cog_pay_t0
	qui sum cog_pay if treated == 0 & (pid_count ==1 | pid_count ==2)
    estadd scalar control_mean = r(mean) 
	
	* Pooled, by Treatment Arm 	
xi: reg cog_pay memory priming thirst `cognitive_controls' if pid_count ==1 | pid_count ==2, cl(pid)  
	eststo cog_pay_t4
	qui sum cog_pay if treated == 0 & (pid_count ==1 | pid_count ==2)
    estadd scalar control_mean = r(mean) 	
	
	* PVT - pooled treated/not
xi: reg pay_pvt1 treated `cognitive_controls' if tag_pid==1  
	eststo pvt_t0
	qui sum pay_pvt1 if treated == 0 & tag_pid==1 
    estadd scalar control_mean = r(mean) 

	* PVT - by treatment
xi: reg pay_pvt1 memory priming thirst `cognitive_controls' if tag_pid==1  
	eststo pvt_t4
	qui sum pay_pvt1 if treated == 0 & tag_pid==1 
    estadd scalar control_mean = r(mean) 

	* Ravens - pooled treated/not
xi: reg pay_rm treated `cognitive_controls' if tag_pid==1  
	eststo ravens_t0
	qui sum pay_rm if treated == 0 & tag_pid==1 
    estadd scalar control_mean = r(mean) 

	* Ravens - by treatment
xi: reg pay_rm memory priming thirst `cognitive_controls' if tag_pid==1  
	eststo ravens_t4
	qui sum pay_rm if treated == 0 & tag_pid==1 
    estadd scalar control_mean = r(mean) 
	
la var pay_rm "RM pay"
la var pay_pvt1 "PVT pay"
la var cog_pay "Pay"

	* Export Table 

#d; 
esttab cog_pay_t0 pvt_t0 ravens_t0 cog_pay_t4 using "$dir_tables/tableS2_cog_performance.tex", replace booktabs 	
	keep(treated memory priming thirst) 
	order(treated memory priming thirst) 
	b(2)
	se label stats(N control_mean, fmt(0 2) labels("Observations" "Control Mean")) star(* 0.10 ** 0.05 *** 0.01) 
	varwidth(25) wrap compress nogap noconstant noomitted numbers nonotes
;#d cr	








	* Table S3 - Efficacy of Treatments 


	* Thirst
xi: regress thirst_ai memory priming thirst `controls' if tag_pid==1  
	eststo thirst
	qui sum thirst_ai if treated == 0 & tag_pid==1 
    estadd scalar control_mean = r(mean) 

	* Thirst, Baseline Control	
gen base_control = thirst_bs

xi: regress thirst_ai memory priming thirst base_control `controls' if tag_pid==1  
	eststo thirst_bs
	qui sum thirst_ai if treated == 0 & tag_pid==1 
    estadd scalar control_mean = r(mean) 

	* Stress 
xi: regress stress_ai memory priming thirst `controls' if tag_pid==1  
	eststo stress
	qui sum stress_ai if treated == 0 & tag_pid==1 
    estadd scalar control_mean = r(mean) 

	* Stress, Baseline Control
replace base_control = . 
replace base_control = stress_bs

xi: regress stress_ai memory priming thirst base_control `controls' if tag_pid==1  
	eststo stress_bs
	qui sum stress_ai if treated == 0 & tag_pid==1 
    estadd scalar control_mean = r(mean) 

lab var base_control "Baseline"

	* Export Table 

#d; 
esttab thirst thirst_bs stress stress_bs using "$dir_tables/tableS3_eff_treatments.tex", replace booktabs 	
	keep(memory priming thirst base_control) 
	order(memory priming thirst base_control) 
	b(2)
	se label stats(N control_mean, fmt(0 2) labels("Observations" "Control Mean")) star(* 0.10 ** 0.05 *** 0.01) 
	varwidth(25) wrap compress nogap noconstant noomitted numbers nonotes
;#d cr	

drop base_control








	* Table S4 - Declines in the Value of Consumption


local controls "score_bs_favoritefood" 

	*** PANEL A - Pooled Treatments ***

	* Pooled treatments, pooled outcomes without control 
xi: reg exp_score treated, cl(pid)
	lincom _b[treated]
	local coef_pool_pool_nc = string(r(estimate),"%3.2f")
	local se_pool_pool_nc = string(r(se),"%3.2f")
	local p_pool_pool_nc = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_pool_pool_nc = string(e(N))
	unique pid if exp_score !=. 
	local ppl_pool_pool_nc = r(sum)
	
	* Pooled treatments, pooled outcomes with control 
xi: reg exp_score treated `controls', cl(pid)
	lincom _b[treated]
	local coef_pool_pool = string(r(estimate),"%3.2f")
	local se_pool_pool = string(r(se),"%3.2f")
	local p_pool_pool = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_pool_pool = string(e(N))
	unique pid if exp_score !=. 
	local ppl_pool_pool = r(sum)
	

sum exp_score if treated == 0 
	local mean_exp_score = string(r(mean), "%3.2f")
	local sd_exp_score = string(r(sd), "%3.2f")

	* Disaggregated outcomes, pooled treatments with control
foreach y in food song video activity {
	xi: reg exp_score treated `controls' if exp_type == "`y'"
		lincom _b[treated]
		local coef_pool_`y' = string(r(estimate),"%3.2f")
		local se_pool_`y' = string(r(se),"%3.2f")
		local p_pool_`y' = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
		
	sum exp_score if exp_type == "`y'" & treated == 0 
		local mean_exp_score_`y' = string(r(mean), "%3.2f")
		local sd_exp_score_`y' = string(r(sd), "%3.2f")
		local obs_pool_`y' = string(e(N))
	}

	*** PANEL B - Disaggregated Treatments ***

	* Disaggregated treatment, pooled outcomes without controls
xi: reg exp_score i.treat, cl(pid)

	test _b[_Itreat_1] = _b[_Itreat_2] = _b[_Itreat_3]
	local f_test_likert_pooled_nc = string(r(p),"%3.2f")
	
	lincom _b[_Itreat_1]
	local coef_prime_pool_nc = string(r(estimate),"%3.2f")
	local se_prime_pool_nc = string(r(se),"%3.2f")
	local p_prime_pool_nc = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	lincom _b[_Itreat_2]
	local coef_mem_pool_nc = string(r(estimate),"%3.2f")
	local se_mem_pool_nc = string(r(se),"%3.2f")
	local p_mem_pool_nc = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	lincom _b[_Itreat_3]
	local coef_thirst_pool_nc = string(r(estimate),"%3.2f")
	local se_thirst_pool_nc = string(r(se),"%3.2f")
	local p_thirst_pool_nc = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))



	* Disaggregated treatment, pooled outcomes with controls
xi: reg exp_score i.treat `controls', cl(pid)

	test _b[_Itreat_1] = _b[_Itreat_2] = _b[_Itreat_3]
	local f_test_likert_pooled = string(r(p),"%3.2f")
	
	lincom _b[_Itreat_1]
	local coef_prime_pool = string(r(estimate),"%3.2f")
	local se_prime_pool = string(r(se),"%3.2f")
	local p_prime_pool = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	lincom _b[_Itreat_2]
	local coef_mem_pool = string(r(estimate),"%3.2f")
	local se_mem_pool = string(r(se),"%3.2f")
	local p_mem_pool = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	lincom _b[_Itreat_3]
	local coef_thirst_pool = string(r(estimate),"%3.2f")
	local se_thirst_pool = string(r(se),"%3.2f")
	local p_thirst_pool = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))

	* Disaggregated treatments, disaggregated outcomes	
foreach y in food song video activity {
	xi: reg exp_score i.treat `controls' if exp_type == "`y'"
	
	test _b[_Itreat_1] = _b[_Itreat_2] = _b[_Itreat_3]
	local f_test_likert_`y' = string(r(p),"%3.2f")
	
		lincom _b[_Itreat_1]
		local coef_prime_`y' = string(r(estimate),"%3.2f")
		local se_prime_`y' = string(r(se),"%3.2f")
		local p_prime_`y' = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
		
		lincom _b[_Itreat_2]
		local coef_mem_`y' = string(r(estimate),"%3.2f")
		local se_mem_`y' = string(r(se),"%3.2f")
		local p_mem_`y' = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
		
		lincom _b[_Itreat_3]
		local coef_thirst_`y' = string(r(estimate),"%3.2f")
		local se_thirst_`y' = string(r(se),"%3.2f")
		local p_thirst_`y' = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	}
	
	* Significance stars
	foreach y in pool food song video activity {
	if `p_pool_`y'' < 0.01 {
		local coef_pool_`y' = "`coef_pool_`y''***"
		}	
	if `p_pool_`y'' < 0.05 & `p_pool_`y'' >= 0.01 {
		local coef_pool_`y' = "`coef_pool_`y''**"
		}	
	if `p_pool_`y'' < 0.1 & `p_pool_`y'' >= 0.05 {
		local coef_pool_`y' = "`coef_pool_`y''*"
		}		
	}
	
	foreach t in mem prime thirst {
	foreach y in pool food song video activity {
		if `p_`t'_`y'' < 0.01 {
			local coef_`t'_`y' = "`coef_`t'_`y''***"
			}
		if `p_`t'_`y'' < 0.05 & `p_`t'_`y'' >= 0.01 {
			local coef_`t'_`y' = "`coef_`t'_`y''**"
			}		
		if `p_`t'_`y'' < 0.1 & `p_`t'_`y'' >= 0.05 {
			local coef_`t'_`y' = "`coef_`t'_`y''*"
			}		
		}
	}
	
	foreach y in pool_pool_nc mem_pool_nc prime_pool_nc thirst_pool_nc {
		if `p_`y'' < 0.01 {
			local coef_`y' = "`coef_`y''***"
			}
		if `p_`y'' < 0.05 & `p_`y'' >= 0.01 {
			local coef_`y' = "`coef_`y''**"
			}		
		if `p_`y'' < 0.1 & `p_`y'' >= 0.05 {
			local coef_`y' = "`coef_`y''*"
			}		
		}
	
	* Export table
	
		file open f using "$dir_tables/tableS4_dec_value_consumption.tex", write replace
			file write f "\begin{tabular}{l*{7}{c}}" _n ///+
			"\toprule" _n ///
			"&\multicolumn{2}{c}{\textbf{Pooled Outcomes}}&\multicolumn{4}{c}{\textbf{Disaggregated Outcomes}}\\" _n ///
			"\cmidrule(lr){2-3}\cmidrule(lr){4-7} & Pooled & Pooled & Food & Song & Video & Game \\" _n ///
			" & (1) & (2) & (3) & (4) & (5) \\" _n ///
			"\midrule" _n ///
			"\textbf{\textit{Panel A: Pooled Treatments}} & & & & &  \\" ///
			"Treated & `coef_pool_pool_nc' & `coef_pool_pool' & `coef_pool_food' & `coef_pool_song' & `coef_pool_video' & `coef_pool_activity' \\" _n ///
			"& (`se_pool_pool_nc') & (`se_pool_pool') & (`se_pool_food') & (`se_pool_song') & (`se_pool_video') & (`se_pool_activity')  \\" _n ///
			"\toprule" _n ///
			"\textbf{\textit{Panel B: Disaggregated Treatments}} & & & & & & \\" ///
			"Memory & `coef_mem_pool_nc' & `coef_mem_pool' & `coef_mem_food' & `coef_mem_song' & `coef_mem_video' & `coef_mem_activity' \\" _n ///
			"& (`se_mem_pool_nc') & (`se_mem_pool') & (`se_mem_food') & (`se_mem_song') & (`se_mem_video') & (`se_mem_activity')  \\" _n ///
			"Financial Stress & `coef_prime_pool_nc' & `coef_prime_pool' & `coef_prime_food' & `coef_prime_song' & `coef_prime_video' & `coef_prime_activity' \\" _n ///
			"& (`se_prime_pool_nc') & (`se_prime_pool') & (`se_prime_food') & (`se_prime_song') & (`se_prime_video') & (`se_prime_activity')  \\" _n ///
			"Thirst & `coef_thirst_pool_nc' & `coef_thirst_pool' & `coef_thirst_food' & `coef_thirst_song' & `coef_thirst_video' & `coef_thirst_activity' \\" _n ///
			"& (`se_thirst_pool_nc') & (`se_thirst_pool') & (`se_thirst_food') & (`se_thirst_song') & (`se_thirst_video') & (`se_thirst_activity')  \\" _n ///
			"\midrule" _n ///
			"Baseline Control & No & Yes & Yes & Yes & Yes & Yes \\" _n ///
			"F-Test & `f_test_likert_pooled_nc' & `f_test_likert_pooled' & `f_test_likert_food' & `f_test_likert_song' & `f_test_likert_video' & `f_test_likert_activity' \\" _n ///
			"Control Mean & `mean_exp_score' & `mean_exp_score' & `mean_exp_score_food' & `mean_exp_score_song' & `mean_exp_score_video' & `mean_exp_score_activity'\\" _n ///
			"Control SD   & `sd_exp_score' & `sd_exp_score'   & `sd_exp_score_food'   & `sd_exp_score_song'   & `sd_exp_score_video'  & `sd_exp_score_activity'  \\" _n ///
			"Observations            & `obs_pool_pool_nc' & `obs_pool_pool'  & `obs_pool_food'  & `obs_pool_song'  & `obs_pool_video' & `obs_pool_activity' \\" _n ///
			"People            & `ppl_pool_pool_nc' & `ppl_pool_pool'  & `obs_pool_food'  & `obs_pool_song'  & `obs_pool_video' & `obs_pool_activity' \\" _n ///			
			"\bottomrule \\" _n ///
			"\end{tabular}" _n
		file close f	
		
		
		
		
		
		
		

	* Table S5 - Heterogeneous Effects by Socio-economic Status


reg exp_score treated z_all_weighted treated_z_all_w if treat == 0 | treat == 1, cl(pid)

	local pids_index_w = e(N_clust)
	local obs_index_w = e(N)

	test _b[treated] + _b[treated_z_all_w] = 0 
	local f_index_w = string(r(p),"%3.2f")

	lincom _b[treated]
	local coef_treat_index_w = string(r(estimate),"%3.2f")
	local se_treat_index_w = string(r(se),"%3.2f")
	local p_treat_index_w = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_treat_index_w = string(e(N))
	
	lincom _b[z_all_weighted]
	local coef_index_index_w = string(r(estimate),"%3.2f")
	local se_index_index_w = string(r(se),"%3.2f")
	local p_index_index_w = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_index_index_w = string(e(N))
	
	lincom _b[treated_z_all_w]
	local coef_int_index_w = string(r(estimate),"%3.2f")
	local se_int_index_w = string(r(se),"%3.2f")
	local p_int_index_w = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_int_index_w = string(e(N))
	

reg exp_score treated##edu_high if treat == 0 | treat == 1, cl(pid) 

	local pids_edu = e(N_clust)
	local obs_edu = e(N)

	test _b[1.treated] + _b[1.treated#1.edu_high] = 0 
	local f_edu = string(r(p),"%3.2f")

	lincom _b[1.treated]
	local coef_treat_edu = string(r(estimate),"%3.2f")
	local se_treat_edu = string(r(se),"%3.2f")
	local p_treat_edu = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_treat_edu = string(e(N))
	
	lincom _b[1.edu_high]
	local coef_edu_edu = string(r(estimate),"%3.2f")
	local se_edu_edu = string(r(se),"%3.2f")
	local p_edu_edu = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_edu_edu = string(e(N))
	
	lincom _b[1.treated#1.edu_high]
	local coef_int_edu = string(r(estimate),"%3.2f")
	local se_int_edu = string(r(se),"%3.2f")
	local p_int_edu = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_int_edu = string(e(N))

reg exp_score treated##earn_weekly_ind_above_p50 if treat == 0 | treat == 1, cl(pid)

	local pids_earn_ind = e(N_clust)
	local obs_earn_ind = e(N)

	test _b[1.treated] + _b[1.treated#1.earn_weekly_ind_above_p50] = 0 
	local f_earn_ind = string(r(p),"%3.2f")

	lincom _b[1.treated]
	local coef_treat_earn_ind = string(r(estimate),"%3.2f")
	local se_treat_earn_ind = string(r(se),"%3.2f")
	local p_treat_earn_ind = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_treat_earn_ind = string(e(N))
	
	lincom _b[1.earn_weekly_ind_above_p50]
	local coef_earn_ind_earn_ind = string(r(estimate),"%3.2f")
	local se_earn_ind_earn_ind = string(r(se),"%3.2f")
	local p_earn_ind_earn_ind = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_earn_ind_earn_ind = string(e(N))
	
	lincom _b[1.treated#1.earn_weekly_ind_above_p50]
	local coef_int_earn_ind = string(r(estimate),"%3.2f")
	local se_int_earn_ind = string(r(se),"%3.2f")
	local p_int_earn_ind = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_int_earn_ind = string(e(N))

reg exp_score treated##earn_weekly_hh_above_p50 if treat == 0 | treat == 1, cl(pid)

	local pids_earn_hh = e(N_clust)
	local obs_earn_hh = e(N)

	test _b[1.treated] + _b[1.treated#1.earn_weekly_hh_above_p50] = 0 
	local f_earn_hh = string(r(p),"%3.2f")

	lincom _b[1.treated]
	local coef_treat_earn_hh = string(r(estimate),"%3.2f")
	local se_treat_earn_hh = string(r(se),"%3.2f")
	local p_treat_earn_hh = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_treat_earn_hh = string(e(N))
	
	lincom _b[1.earn_weekly_hh_above_p50]
	local coef_earn_hh_earn_hh = string(r(estimate),"%3.2f")
	local se_earn_hh_earn_hh = string(r(se),"%3.2f")
	local p_earn_hh_earn_hh = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_earn_hh_earn_hh = string(e(N))
	
	lincom _b[1.treated#1.earn_weekly_hh_above_p50]
	local coef_int_earn_hh = string(r(estimate),"%3.2f")
	local se_int_earn_hh = string(r(se),"%3.2f")
	local p_int_earn_hh = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_int_earn_hh = string(e(N))

reg exp_score treated##emp_full_time if treat == 0 | treat == 1, cl(pid)

	local pids_emp = e(N_clust)
	local obs_emp = e(N)
	
	test _b[1.treated] + _b[1.treated#1.emp_full_time] = 0 
	local f_emp = string(r(p),"%3.2f")

lincom _b[1.treated]
	local coef_treat_emp = string(r(estimate),"%3.2f")
	local se_treat_emp = string(r(se),"%3.2f")
	local p_treat_emp = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_treat_emp = string(e(N))
	
	lincom _b[1.emp_full_time]
	local coef_emp_emp = string(r(estimate),"%3.2f")
	local se_emp_emp = string(r(se),"%3.2f")
	local p_emp_emp = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_emp_emp = string(e(N))
	
	lincom _b[1.treated#1.emp_full_time]
	local coef_int_emp = string(r(estimate),"%3.2f")
	local se_int_emp = string(r(se),"%3.2f")
	local p_int_emp = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_int_emp = string(e(N))

reg exp_score treated##appliance_tv if treat == 0 | treat == 1, cl(pid)

	local pids_app = e(N_clust)
	local obs_app = e(N)

	test _b[1.treated] + _b[1.treated#1.appliance_tv] = 0 
	local f_app = string(r(p),"%3.2f")

	lincom _b[1.treated]
	local coef_treat_app = string(r(estimate),"%3.2f")
	local se_treat_app = string(r(se),"%3.2f")
	local p_treat_app = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_treat_app = string(e(N))
	
	lincom _b[1.appliance_tv]
	local coef_app_app = string(r(estimate),"%3.2f")
	local se_app_app = string(r(se),"%3.2f")
	local p_app_app = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_app_app = string(e(N))
	
	lincom _b[1.treated#1.appliance_tv]
	local coef_int_app = string(r(estimate),"%3.2f")
	local se_int_app = string(r(se),"%3.2f")
	local p_int_app = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_int_app = string(e(N))
	
	
	* Significance stars
	
foreach t in treat int {
	foreach y in edu earn_ind earn_hh emp app index_w{
		if `p_`t'_`y'' < 0.01 {
			local coef_`t'_`y' = "`coef_`t'_`y''***"
			}
		if `p_`t'_`y'' < 0.05 & `p_`t'_`y'' >= 0.01 {
			local coef_`t'_`y' = "`coef_`t'_`y''**"
			}		
		if `p_`t'_`y'' < 0.1 & `p_`t'_`y'' >= 0.05 {
			local coef_`t'_`y' = "`coef_`t'_`y''*"
			}		
		}
	}
	
foreach y in edu_edu earn_ind_earn_ind earn_hh_earn_hh emp_emp app_app index_index_w {
	if `p_`y'' < 0.01 {
		local coef_`y' = "`coef_`y''***"
		}	
	if `p_`y'' < 0.05 & `p_`y'' >= 0.01 {
		local coef_`y' = "`coef_`y''**"
		}	
	if `p_`y'' < 0.1 & `p_`y'' >= 0.05 {
		local coef_`y' = "`coef_`y''*"
		}		
	}
	
	* Export table 
	
	file open f using "$dir_tables/tableS5_hte_by_SES.tex", write replace
			file write f "\begin{tabular}{l*{6}{c}}" _n ///+
			"\toprule" _n ///
			"&\multicolumn{5}{c}{\textbf{Dependent Variable: Likert-scale Score}} \\" _n ///
			"\midrule" _n ///
			"&\multicolumn{5}{c}{\textbf{Heterogeneity Covariate}} \\" _n ///
			"\cmidrule(lr){2-6} & Anderson Index & Employment & Income & Education & TV Ownership \\" _n ///
			" & (1) & (2) & (3) & (4) & (5) \\" _n ///
			"\midrule" _n ///
			"Financial Stress & `coef_treat_index_w' & `coef_treat_emp' & `coef_treat_earn_ind' & `coef_treat_edu' & `coef_treat_app' \\" _n ///
			" & (`se_treat_index_w') & (`se_treat_emp') & (`se_treat_earn_ind') & (`se_treat_edu') & (`se_treat_app') \\" _n ///
			"Hte. Covariate & `coef_index_index_w' & `coef_emp_emp' & `coef_earn_ind_earn_ind' & `coef_edu_edu' & `coef_app_app' \\" _n ///
			" & (`se_index_index_w') & (`se_emp_emp') & (`se_earn_ind_earn_ind') & (`se_edu_edu') & (`se_app_app') \\" _n ///
			"Fin. Stress * Hte. Covariate & `coef_int_index_w' & `coef_int_emp' & `coef_int_earn_ind' & `coef_int_edu' & `coef_int_app' \\" _n ///
			" & (`se_int_index_w') & (`se_int_emp') & (`se_int_earn_ind') & (`se_int_edu') & (`se_int_app') \\" _n ///
			"\midrule" _n ///
			"F-Test & `f_index_w' & `f_emp' & `f_earn_ind' & `f_edu' & `f_app' \\" _n ///
			"Observations & `obs_index_w' & `obs_emp' & `obs_earn_ind' & `obs_edu' & `obs_app' \\" _n ///
			"People & `pids_index_w' & `pids_emp' & `pids_earn_ind' & `pids_edu' & `pids_app' \\" _n ///			
			"\bottomrule \\" _n ///
			"\end{tabular}" _n
		file close f	

		
		

		
		
		
	* Table S6 - Robustness of Declines in Consumption Value


local controls "score_bs_favoritefood" 


	* Pooled treatments, pooled outcomes
xi: reg exp_score treated `controls', cl(pid)
	lincom _b[treated]
	local coef_pool_pool = string(r(estimate),"%3.2f")
	local se_pool_pool = string(r(se),"%4.3f")
	local p_pool_pool = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_pool_pool = string(e(N))
	unique pid if exp_score !=. 
	local ppl_pool_pool = r(sum)
	
	* No controls 
	reg exp_score treated, cl(pid)
	lincom _b[treated]
	local coef_pool_pool_nc = string(r(estimate),"%3.2f")
	local se_pool_pool_nc = string(r(se),"%4.3f")
	local p_pool_pool_nc = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_pool_pool_nc = string(e(N))
	unique pid if exp_score !=. 
	local ppl_pool_pool_nc = r(sum)

xi: reg exp_score treated `controls' age, cl(pid)
	lincom _b[treated]
	local coef_2 = string(r(estimate),"%3.2f")
	local se_2 = string(r(se),"%4.3f")
	local p_2 = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_2 = string(e(N))
	unique pid if exp_score !=. 
	local ppl_2 = r(sum)
	
	
xi: reg exp_score treated `controls' age married, cl(pid)
	lincom _b[treated]
	local coef_3 = string(r(estimate),"%3.2f")
	local se_3 = string(r(se),"%4.3f")
	local p_3 = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_3 = string(e(N))
	unique pid if exp_score !=. 
	local ppl_3 = r(sum)

xi: reg exp_score treated `controls' age married employed_outhh, cl(pid)
	lincom _b[treated]
	local coef_4 = string(r(estimate),"%3.2f")
	local se_4 = string(r(se),"%4.3f")
	local p_4 = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_4 = string(e(N))
	unique pid if exp_score !=. 
	local ppl_4 = r(sum)
	
xi: reg exp_score treated `controls' age married employed_outhh earn_daily_hh_p95, cl(pid)
	lincom _b[treated]
	local coef_5 = string(r(estimate),"%3.2f")
	local se_5 = string(r(se),"%4.3f")
	local p_5 = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_5 = string(e(N))
	unique pid if exp_score !=. 
	local ppl_5 = r(sum)
	
xi: reg exp_score treated `controls' age married employed_outhh earn_daily_hh_p95 thirst_bs, cl(pid)
	lincom _b[treated]
	local coef_6 = string(r(estimate),"%3.2f")
	local se_6 = string(r(se),"%4.3f")
	local p_6 = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_6 = string(e(N))
	unique pid if exp_score !=. 
	local ppl_6 = r(sum)

	// controls: only sig at 5% 
	
xi: reg exp_score treated `controls' age earn_daily_hh thirst_bs, cl(pid)
	lincom _b[treated]
	local coef_7 = string(r(estimate),"%3.2f")
	local se_7 = string(r(se),"%4.3f")
	local p_7 = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_7 = string(e(N))
	unique pid if exp_score !=. 
	local ppl_7 = r(sum)
	
	* Significance stars
foreach y in _pool_pool _2 _3 _4 _5 _6 _7 {
	if `p`y'' < 0.01 {
		local coef`y' = "`coef`y''***"
		}	
	if `p`y'' < 0.05 & `p`y'' >= 0.01 {
		local coef`y' = "`coef`y''**"
		}	
	if `p`y'' < 0.1 & `p`y'' >= 0.05 {
		local coef`y' = "`coef`y''*"
		}		
	}
	
	* Disaggregated treatment, pooled outcomes  
	xi: reg exp_score i.treat `controls', cl(pid)

	test _b[_Itreat_1] = _b[_Itreat_2] = _b[_Itreat_3]
	local f_pool = string(r(p),"%3.2f")
	
	lincom _b[_Itreat_1]
	local coef_prime_pool = string(r(estimate),"%3.2f")
	local se_prime_pool = string(r(se),"%3.2f")
	local p_prime_pool = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	lincom _b[_Itreat_2]
	local coef_mem_pool = string(r(estimate),"%3.2f")
	local se_mem_pool = string(r(se),"%3.2f")
	local p_mem_pool = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	lincom _b[_Itreat_3]
	local coef_thirst_pool = string(r(estimate),"%3.2f")
	local se_thirst_pool = string(r(se),"%3.2f")
	local p_thirst_pool = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	
	* Disaggregated treatment, pooled outcomes, no controls  
	xi: reg exp_score i.treat, cl(pid)

	test _b[_Itreat_1] = _b[_Itreat_2] = _b[_Itreat_3]
	local f_pool_nc = string(r(p),"%3.2f")
	
	lincom _b[_Itreat_1]
	local coef_prime_pool_nc = string(r(estimate),"%3.2f")
	local se_prime_pool_nc = string(r(se),"%3.2f")
	local p_prime_pool_nc = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	lincom _b[_Itreat_2]
	local coef_mem_pool_nc = string(r(estimate),"%3.2f")
	local se_mem_pool_nc = string(r(se),"%3.2f")
	local p_mem_pool_nc = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	lincom _b[_Itreat_3]
	local coef_thirst_pool_nc = string(r(estimate),"%3.2f")
	local se_thirst_pool_nc = string(r(se),"%3.2f")
	local p_thirst_pool_nc = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	* extra controls  
	local c2 "age" 
	local c3 "age married"
	local c4 "age married employed_outhh" 
	local c5 "age married employed_outhh earn_daily_hh_p95 "
	local c6 "age married employed_outhh earn_daily_hh_p95 thirst_bs" 
	local c7 "age earn_daily_hh thirst_bs" 

	xi: reg exp_score i.treat `controls' `c2', cl(pid)

	test _b[_Itreat_1] = _b[_Itreat_2] = _b[_Itreat_3]
	local f_2 = string(r(p),"%3.2f")
	
	lincom _b[_Itreat_1]
	local coef_prime_2 = string(r(estimate),"%3.2f")
	local se_prime_2 = string(r(se),"%3.2f")
	local p_prime_2 = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	lincom _b[_Itreat_2]
	local coef_mem_2 = string(r(estimate),"%3.2f")
	local se_mem_2 = string(r(se),"%3.2f")
	local p_mem_2 = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	lincom _b[_Itreat_3]
	local coef_thirst_2 = string(r(estimate),"%3.2f")
	local se_thirst_2 = string(r(se),"%3.2f")
	local p_thirst_2 = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))

	// 3

	xi: reg exp_score i.treat `controls' `c3', cl(pid)

	test _b[_Itreat_1] = _b[_Itreat_2] = _b[_Itreat_3]
	local f_3 = string(r(p),"%3.2f")
	
	lincom _b[_Itreat_1]
	local coef_prime_3 = string(r(estimate),"%3.2f")
	local se_prime_3 = string(r(se),"%3.2f")
	local p_prime_3 = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	lincom _b[_Itreat_2]
	local coef_mem_3 = string(r(estimate),"%3.2f")
	local se_mem_3 = string(r(se),"%3.2f")
	local p_mem_3 = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	lincom _b[_Itreat_3]
	local coef_thirst_3 = string(r(estimate),"%3.2f")
	local se_thirst_3 = string(r(se),"%3.2f")
	local p_thirst_3 = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	// 4

	xi: reg exp_score i.treat `controls' `c4', cl(pid)

	test _b[_Itreat_1] = _b[_Itreat_2] = _b[_Itreat_3]
	local f_4 = string(r(p),"%3.2f")
	
	lincom _b[_Itreat_1]
	local coef_prime_4 = string(r(estimate),"%3.2f")
	local se_prime_4 = string(r(se),"%3.2f")
	local p_prime_4 = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	lincom _b[_Itreat_2]
	local coef_mem_4 = string(r(estimate),"%3.2f")
	local se_mem_4 = string(r(se),"%3.2f")
	local p_mem_4 = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	lincom _b[_Itreat_3]
	local coef_thirst_4 = string(r(estimate),"%3.2f")
	local se_thirst_4 = string(r(se),"%3.2f")
	local p_thirst_4 = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	// 5

	xi: reg exp_score i.treat `controls' `c5', cl(pid)

	test _b[_Itreat_1] = _b[_Itreat_2] = _b[_Itreat_3]
	local f_5 = string(r(p),"%3.2f")
	
	lincom _b[_Itreat_1]
	local coef_prime_5 = string(r(estimate),"%3.2f")
	local se_prime_5 = string(r(se),"%3.2f")
	local p_prime_5 = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	lincom _b[_Itreat_2]
	local coef_mem_5 = string(r(estimate),"%3.2f")
	local se_mem_5 = string(r(se),"%3.2f")
	local p_mem_5 = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	lincom _b[_Itreat_3]
	local coef_thirst_5 = string(r(estimate),"%3.2f")
	local se_thirst_5 = string(r(se),"%3.2f")
	local p_thirst_5 = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	// 6

	xi: reg exp_score i.treat `controls' `c6', cl(pid)

	test _b[_Itreat_1] = _b[_Itreat_2] = _b[_Itreat_3]
	local f_6 = string(r(p),"%3.2f")
	
	lincom _b[_Itreat_1]
	local coef_prime_6 = string(r(estimate),"%3.2f")
	local se_prime_6 = string(r(se),"%3.2f")
	local p_prime_6 = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	lincom _b[_Itreat_2]
	local coef_mem_6 = string(r(estimate),"%3.2f")
	local se_mem_6 = string(r(se),"%3.2f")
	local p_mem_6 = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	lincom _b[_Itreat_3]
	local coef_thirst_6 = string(r(estimate),"%3.2f")
	local se_thirst_6 = string(r(se),"%3.2f")
	local p_thirst_6 = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	// 7 

	xi: reg exp_score i.treat `controls' `c7', cl(pid)

	test _b[_Itreat_1] = _b[_Itreat_2] = _b[_Itreat_3]
	local f_7 = string(r(p),"%3.2f")
	
	lincom _b[_Itreat_1]
	local coef_prime_7 = string(r(estimate),"%3.2f")
	local se_prime_7 = string(r(se),"%3.2f")
	local p_prime_7 = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	lincom _b[_Itreat_2]
	local coef_mem_7 = string(r(estimate),"%3.2f")
	local se_mem_7 = string(r(se),"%3.2f")
	local p_mem_7 = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	lincom _b[_Itreat_3]
	local coef_thirst_7 = string(r(estimate),"%3.2f")
	local se_thirst_7 = string(r(se),"%3.2f")
	local p_thirst_7 = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	

	* Significance stars
	
foreach y in prime mem thirst {
foreach x in 2 3 4 5 6 7 pool {
	if `p_`y'_`x'' < 0.01 {
		local coef_`y'_`x' = "`coef_`y'_`x''***"
		}	
	if `p_`y'_`x'' < 0.05 & `p_`y'_`x'' >= 0.01 {
		local coef_`y'_`x' = "`coef_`y'_`x''**"
		}	
	if `p_`y'_`x'' < 0.1 & `p_`y'_`x'' >= 0.05 {
		local coef_`y'_`x' = "`coef_`y'_`x''*"
		}		
	}
	}
	
	foreach y in pool_pool_nc thirst_pool_nc mem_pool_nc prime_pool_nc {
	if `p_`y'' < 0.01 {
		local coef_`y' = "`coef_`y''***"
		}	
	if `p_`y'' < 0.05 & `p_`y'' >= 0.01 {
		local coef_`y' = "`coef_`y''**"
		}	
	if `p_`y'' < 0.1 & `p_`y'' >= 0.05 {
		local coef_`y' = "`coef_`y''*"
		}		
	}

		
	*Export table
	
		file open f using "$dir_tables/tableS6_robustness.tex", write replace
			file write f "\begin{tabular}{l*{5}{c}}" _n ///+
			"\toprule" _n ///
			"\cmidrule(lr){2-5}&\multicolumn{4}{c}{\textbf{Dep. Variable: Likert Score}}\\" _n ///
			" & (1) & (2) & (3) \& (4)\" _n ///
			"\midrule" _n ///
			"&\multicolumn{4}{c}{\textbf{Panel A: Pooled Treatments}} \\" _n ///
			"Treated & `coef_pool_pool_nc' & `coef_pool_pool' & `coef_7' & `coef_6'\\" _n ///
			"& (`se_pool_pool_nc') & (`se_pool_pool') & (`se_7') & (`se_6') \\" _n ///
			"\midrule" _n ///
			"&\multicolumn{4}{c}{\textbf{Panel B: Disagg. Treatments}} \\" _n ///
			"Memory & `coef_mem_pool_nc' & `coef_mem_pool' & `coef_mem_7' & `coef_mem_6' \\" _n ///
			" & (`se_mem_pool_nc') & (`se_mem_pool') & (`se_mem_7') & (`se_mem_6') \\" _n ///
			"Financial Stress & `coef_prime_pool_nc' & `coef_prime_pool' & `coef_prime_7' & `coef_prime_6' \\" _n ///
			"& (`se_prime_pool_nc') & (`se_prime_pool') & (`se_prime_7') & (`se_prime_6') \\" _n ///
			"Thirst & `coef_thirst_pool_nc' & `coef_thirst_pool' & `coef_thirst_7' & `coef_thirst_6' \\" _n ///
			" & (`se_thirst_pool_nc') & (`se_thirst_pool') & (`se_thirst_7') & (`se_thirst_6') \\" _n ///
			"\toprule" _n ///
			"Baseline Likert Rating & No & Yes & Yes & Yes   \\" _n ///
			"Age & No & No & Yes & Yes   \\" _n ///
			"Income & No & No & Yes & Yes  \\" _n ///
			"Baseline Thirst & No & No & Yes & Yes  \\" _n ///
			"Marital Status & No & No & No & Yes  \\" _n ///
			"Employment & No & No & No & Yes  \\" _n ///
			"\midrule" _n ///
			" F-test & `f_pool_nc' & `f_pool' & `f_7' & `f_6' \\" _n /// 
			"Observations & `obs_pool_pool_nc' & `obs_pool_pool' & `obs_7' & `obs_6' \\" _n ///
			"People & `ppl_pool_pool_nc' & `ppl_pool_pool' & `ppl_7' & `ppl_6' \\" _n ///			
			"\bottomrule \\" _n ///
			"\end{tabular}" _n
		file close f		



	* Table S7 - Willingness to Pay 

	*** PANEL A - pooled treatments ***
	
	* Pooled treatments, pooled outcomes
xi: reg wtp_value treated `controls', cl(pid)
	lincom _b[treated]
	local coef_pool_pool_bdm = string(r(estimate),"%3.2f")
	local se_pool_pool_bdm = string(r(se),"%3.2f")
	local p_pool_pool_bdm = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	local obs_pool_pool_bdm = string(e(N))
	unique pid if wtp_value!=. 
	local ppl_pool_pool_bdm = r(sum)
	

sum wtp_value if treated == 0 
	local mean_wtp_value = string(r(mean), "%3.2f")
	local sd_wtp_value = string(r(sd), "%3.2f")

	* Disaggregated treatments, pooled outcomes
foreach y in food song video activity {
	xi: reg wtp_value treated `controls' if exp_type == "`y'"
		lincom _b[treated]
		local coef_pool_`y'_bdm = string(r(estimate),"%3.2f")
		local se_pool_`y'_bdm = string(r(se),"%3.2f")
		local p_pool_`y'_bdm = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
		
	sum wtp_value if exp_type == "`y'" & treated == 0 
		local mean_wtp_value_`y' = string(r(mean), "%3.2f")
		local sd_wtp_value_`y' = string(r(sd), "%3.2f")
		local obs_pool_`y'_bdm = string(e(N))
	}

	*** PANEL B - disagg treatments ***
	
	* Disaggregated treatment, pooled outcomes  
xi: reg wtp_value i.treat `controls', cl(pid)
	lincom _b[_Itreat_1]
	local coef_prime_pool_bdm = string(r(estimate),"%3.2f")
	local se_prime_pool_bdm = string(r(se),"%3.2f")
	local p_prime_pool_bdm = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	lincom _b[_Itreat_2]
	local coef_mem_pool_bdm = string(r(estimate),"%3.2f")
	local se_mem_pool_bdm = string(r(se),"%3.2f")
	local p_mem_pool_bdm = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	
	lincom _b[_Itreat_3]
	local coef_thirst_pool_bdm = string(r(estimate),"%3.2f")
	local se_thirst_pool_bdm = string(r(se),"%3.2f")
	local p_thirst_pool_bdm = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))

	* Disaggregated treatments, disagg outcomes	
foreach y in food song video activity {
	xi: reg wtp_value i.treat `controls' if exp_type == "`y'"
		lincom _b[_Itreat_1]
		local coef_prime_`y'_bdm = string(r(estimate),"%3.2f")
		local se_prime_`y'_bdm = string(r(se),"%3.2f")
		local p_prime_`y'_bdm = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
		
		lincom _b[_Itreat_2]
		local coef_mem_`y'_bdm = string(r(estimate),"%3.2f")
		local se_mem_`y'_bdm = string(r(se),"%3.2f")
		local p_mem_`y'_bdm = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
		
		lincom _b[_Itreat_3]
		local coef_thirst_`y'_bdm = string(r(estimate),"%3.2f")
		local se_thirst_`y'_bdm = string(r(se),"%3.2f")
		local p_thirst_`y'_bdm = (2 * ttail(e(df_r), abs(r(estimate)/r(se))))
	}
	
	* Significance stars
foreach y in pool food song video activity {
	if `p_pool_`y'_bdm' < 0.01 {
		local coef_pool_`y'_bdm = "`coef_pool_`y'_bdm'***"
		}	
	if `p_pool_`y'_bdm' < 0.05 & `p_pool_`y'_bdm' >= 0.01 {
		local coef_pool_`y'_bdm = "`coef_pool_`y'_bdm'**"
		}	
	if `p_pool_`y'_bdm' < 0.1 & `p_pool_`y'_bdm' >= 0.05 {
		local coef_pool_`y'_bdm = "`coef_pool_`y'_bdm'*"
		}		
	}
	
foreach t in mem prime thirst {
	foreach y in pool food song video activity {
		if `p_`t'_`y'_bdm' < 0.01 {
			local coef_`t'_`y'_bdm = "`coef_`t'_`y'_bdm'***"
			}
		if `p_`t'_`y'_bdm' < 0.05 & `p_`t'_`y'_bdm' >= 0.01 {
			local coef_`t'_`y'_bdm = "`coef_`t'_`y'_bdm'**"
			}		
		if `p_`t'_`y'_bdm' < 0.1 & `p_`t'_`y'_bdm' >= 0.05 {
			local coef_`t'_`y'_bdm = "`coef_`t'_`y'_bdm'*"
			}		
		}
	}
		

	* Export table
	
		file open f using "$dir_tables/tableS7_WTP.tex", write replace
			file write f "\begin{tabular}{l*{6}{c}}" _n ///+
			"\toprule" _n ///
			"&\multicolumn{1}{c}{\textbf{Pooled Outcomes}}&\multicolumn{4}{c}{\textbf{Disaggregated Outcomes}}\\" _n ///
			"\cmidrule(lr){2-2}\cmidrule(lr){3-6} & Pooled & Food & Song & Video & Game \\" _n ///
			" & (1) & (2) & (3) & (4) & (5) \\" _n ///
			"\midrule" _n ///
			"\textbf{\textit{Panel A: Pooled Treatments}} & & & & &  \\" ///
			"Treated & `coef_pool_pool_bdm' & `coef_pool_food_bdm' & `coef_pool_song_bdm' & `coef_pool_video_bdm' & `coef_pool_activity_bdm' \\" _n ///
			"& (`se_pool_pool_bdm') & (`se_pool_food_bdm') & (`se_pool_song_bdm') & (`se_pool_video_bdm') & (`se_pool_activity_bdm')  \\" _n ///
			"\midrule" _n ///
			"\textbf{\textit{Panel B: Disaggregated Treatments}} & & & & & & \\" ///
			"Memory & `coef_mem_pool_bdm' & `coef_mem_food_bdm' & `coef_mem_song_bdm' & `coef_mem_video_bdm' & `coef_mem_activity_bdm' \\" _n ///
			"& (`se_mem_pool_bdm') & (`se_mem_food_bdm') & (`se_mem_song_bdm') & (`se_mem_video_bdm') & (`se_mem_activity_bdm')  \\" _n ///
			"Financial Stress & `coef_prime_pool_bdm' & `coef_prime_food_bdm' & `coef_prime_song_bdm' & `coef_prime_video_bdm' & `coef_prime_activity_bdm' \\" _n ///
			"& (`se_prime_pool_bdm') & (`se_prime_food_bdm') & (`se_prime_song_bdm') & (`se_prime_video_bdm') & (`se_prime_activity_bdm')  \\" _n ///
			"Thirst & `coef_thirst_pool_bdm' & `coef_thirst_food_bdm' & `coef_thirst_song_bdm' & `coef_thirst_video_bdm' & `coef_thirst_activity_bdm' \\" _n ///
			"& (`se_thirst_pool_bdm') & (`se_thirst_food_bdm') & (`se_thirst_song_bdm') & (`se_thirst_video_bdm') & (`se_thirst_activity_bdm')  \\" _n ///
			"\midrule" _n ///
			"Control Mean & `mean_wtp_value' & `mean_wtp_value_food' & `mean_wtp_value_song' & `mean_wtp_value_video' & `mean_wtp_value_activity'\\" _n ///
			"Control SD   & `sd_wtp_value'   & `sd_wtp_value_food'   & `sd_wtp_value_song'   & `sd_wtp_value_video'  & `sd_wtp_value_activity'  \\" _n ///
			"Observations & `obs_pool_pool_bdm'  & `obs_pool_food_bdm'  & `obs_pool_song_bdm'  & `obs_pool_video_bdm' & `obs_pool_activity_bdm' \\" _n ///
			"People & `ppl_pool_pool_bdm'  & `obs_pool_food_bdm'  & `obs_pool_song_bdm'  & `obs_pool_video_bdm' & `obs_pool_activity_bdm' \\" _n ///
			"\bottomrule \\" _n ///
			"\end{tabular}" _n
		file close f	
	

	

